

Universidad Autónoma de la Ciudad de México

Licenciatura en Ingeniería de Software

Arquitectura de Computadoras

Práctica 03

Factorial Recursivo en Ensamblador x86

Alumno: Edson Joel Carrera Avila

Grupo: 302

Profesor: Prof. Mtro. Omar Nieto Crisóstomo

Fecha: May 23, 2026

Contents

1	Introducción	2
2	Organización del programa	2
3	Código fuente	2
3.1	Programa principal (main.cpp)	2
3.2	Función ensambladora (factorial.asm)	3
4	Cómo funciona la recursión	4
5	La Unidad de Punto Flotante (FPU)	5
6	Ejemplo de ejecución	6
7	Repositorio GitHub	7
8	Conclusiones	7

1 Introducción

Esta práctica consiste en calcular el **factorial** de un número usando dos lenguajes al mismo tiempo: **C++** (entrada y salida de datos) y **ensamblador x86** (operación matemática). La idea es ver cómo un programa de alto nivel puede delegar trabajo a código ensamblador, que está más cerca del hardware.

El factorial de un número es multiplicar ese número por todos los enteros positivos menores que él. Por ejemplo:

$$5! = 5 \times 4 \times 3 \times 2 \times 1 = 120 \quad (1)$$

Matemáticamente, se puede definir de forma **recursiva** (es decir, usando la misma operación pero con un valor más pequeño cada vez):

$$x! = \begin{cases} 1 & \text{si } x \leq 1 \quad (\text{caso base}) \\ x \cdot (x-1)! & \text{si } x > 1 \quad (\text{paso recursivo}) \end{cases} \quad (2)$$

2 Organización del programa

El proyecto tiene dos archivos que trabajan juntos:

Table 1: Archivos del proyecto

Archivo	Qué hace
<code>main.cpp</code>	Le pide un número al usuario, llama a la función ensambladora y muestra el resultado.
<code>factorial.asm</code>	Contiene el cálculo del factorial escrito en ensamblador con instrucciones FPU.

El flujo del programa es sencillo:

1. El usuario escribe un número en la consola.
2. C++ recibe ese número y lo pasa a la función ensambladora.
3. La función ensambladora calcula el factorial de forma recursiva.
4. C++ toma el resultado y lo imprime en pantalla.

3 Código fuente

3.1 Programa principal (main.cpp)

El código en C++ hace tres cosas: declarar que la función `Mi_fact` existe en otro archivo, pedirle un número al usuario, y mostrar el resultado.

```

1 #include <iostream>
2 using namespace std;
3
4 // Le decimos a C++ que esta funcion esta definida en el archivo .asm
5 extern "C" double Mi_fact(double x);
6
7 int main() {
8     double numero;
9     cout << "--- Calculadora de Factorial (ASM Recursivo) ---" << endl;
10    cout << "Introduce un numero: ";
11    cin >> numero;
12
13    double resultado = Mi_fact(numero);
14    cout << "El factorial de " << numero
15         << " es: " << resultado << endl;
16    return 0;
17 }

```

Listing 1: main.cpp — Entrada, salida y llamada a la función ASM

La línea **extern "C"** es el puente entre C++ y ASM. Le dice al compilador que no modifique el nombre de la función al momento de compilar, para que ambos archivos puedan encontrarse sin problemas de decoración de nombres.

3.2 Función ensambladora (factorial.asm)

Aquí está el núcleo del programa. La función recibe *x*, comprueba si es 1 o menos (caso base), y si no, se llama a sí misma con *x - 1* y multiplica el resultado por *x*.

```

1 .586
2 .model flat, c
3
4 .code
5
6 Mi_fact PROC
7
8     ; Preparar el marco de pila
9     PUSH    EBP
10    MOV     EBP, ESP
11    SUB     ESP, 8           ; Reservar 8 bytes para la variable local
12                          (x - 1.0)
13
14    FLD     QWORD PTR [EBP+8] ; Cargar el parametro x en la pila FPU ->
15                          ST(0) = x
16
17    ; Comparar x con 1.0
18    FLD1
19    FCOMIP  ST(0), ST(1)      ; ST(0) = 1.0, ST(1) = x
20                          ; Comparar 1.0 con x; actualiza flags de
21                          CPU
22    JAE     caso_base        ; Si 1.0 >= x (es decir x <= 1), ir al
23                          caso base
24
25    ; x > 1: calcular x * Mi_fact(x - 1)

```

```

21  FLD1                ; ST(0) = 1.0, ST(1) = x
22  FSUBP    ST(1), ST(0) ; ST(0) = x - 1.0
23  FSTP     QWORD PTR [EBP-8] ; Guardar (x - 1.0) en variable local y
    vaciar FPU
24
25  PUSH     DWORD PTR [EBP-4] ; Pasar (x-1.0) parte alta a la pila
26  PUSH     DWORD PTR [EBP-8] ; Pasar (x-1.0) parte baja a la pila
27  CALL     Mi_fact          ; Llamada recursiva; resultado queda en
    ST(0)
28  ADD      ESP, 8          ; Limpiar el argumento de la pila
29
30  FMUL     QWORD PTR [EBP+8] ; ST(0) = Mi_fact(x-1) * x
31  JMP      fin
32
33  caso_base:
34  FSTP     ST(0)          ; Descartar x de la FPU
35  FLD1                ; Retornar 1.0 en ST(0)
36
37  fin:
38  MOV      ESP, EBP
39  POP      EBP
40  RET
41
42  Mi_fact  ENDP
43  END


```

Listing 2: factorial.asm — Implementación recursiva con FPU

4 Cómo funciona la recursión

Cada vez que la función se llama a sí misma, guarda su estado actual en la **pila de memoria**, resuelve el subproblema más pequeño y, al regresar, retoma donde se quedó.

Para `Mi_fact(4.0)`, la cadena de llamadas es:

Figure 1: Cadena de llamadas recursivas para `Mi_fact(4.0) = 24`

Cada nivel espera a que el nivel inferior termine, multiplica el resultado por su x y lo regresa al nivel superior.




Figure 2: Ventana Call Stack de Visual Studio mostrando las llamadas recursivas apiladas de `Mi_fact`.

5 La Unidad de Punto Flotante (FPU)

El número se maneja como `double` (número con decimales de 64 bits) porque así lo requiere la interfaz con C++. Para operar con este tipo de dato, el ensamblador usa la **FPU**, un componente del procesador dedicado a la aritmética con decimales. La FPU tiene su propia pequeña pila interna de registros llamados `ST(0)` a `ST(7)`.

Las instrucciones FPU que usa esta práctica son:

Table 2: Instrucciones FPU utilizadas

Instrucción	Qué hace en términos simples
FLD	Toma un número de la memoria y lo coloca en la cima de la pila FPU (ST(0)).
FLD1	Coloca directamente el valor 1.0 en la cima de la pila FPU.
FCOMIP	Compara ST(0) con otro registro FPU, actualiza los flags de la CPU y saca ST(0) de la pila.
FSUBP	Resta los dos valores del tope de la pila FPU y saca el operando fuente.
FSTP	Saca el valor del tope de la pila FPU y lo guarda en memoria.
FMUL	Multiplica la cima de la pila FPU por un valor en memoria.





Figure 3: Ventana con los registros FPU ST(0)–ST(7). En este caso $ST(0) = 1.2 \times 10^2 = 120$

6 Ejemplo de ejecución

Al correr el programa, la consola muestra:

```
--- Calculadora de Factorial (ASM Recursivo) ---
Introduce un numero: 6
El factorial de 6 es: 720
```



```
Consola de depuración de Mi x + v
--- Calculadora de Factorial (ASM Recursivo) ---
Introduce un numero: 6
El factorial de 6 es: 720

C:\Users\ING EJCA\Desktop\Practica03_FactorialRecursivo\Debug\Practica03_FactorialRecursivo.exe (proceso 19052) se cerró con el código 0 (0x0).
Para cerrar automáticamente la consola cuando se detiene la depuración, habilite Herramientas ->Opciones ->Depuración ->Cerrar la consola automáticamente al detenerse la depuración.
Presione cualquier tecla para cerrar esta ventana. . .]
```

Figure 4: Consola de ejecución: el usuario introduce 6 y el programa devuelve 720.

7 Repositorio GitHub

https://github.com/7mo-ArquitecturaComputadoras/Practica03_FactorialRecursivo

8 Conclusiones

- Fue posible combinar C++ con ensamblador de forma sencilla gracias a `extern "C"`, que actúa como puente entre los dos lenguajes al evitar la decoración de nombres del compilador.
- La recursión en ensamblador funciona igual que en cualquier lenguaje de alto nivel: cada llamada guarda su estado en la pila, espera el resultado del nivel inferior y continúa. La diferencia es que en ensamblador todo ese manejo se hace instrucción por instrucción, de forma manual.
- El uso de la FPU permite trabajar con números de 64 bits (`double`), lo que hace a la función compatible con el tipo `double` de C++ y garantiza precisión suficiente para los factoriales de números pequeños.
- La instrucción `FCOMIP` es clave: compara dos valores en la pila FPU y actualiza directamente los flags enteros de la CPU, permitiendo usar saltos condicionales (`JAE`) sin necesidad de instrucciones adicionales de transferencia.
- Esta práctica ayuda a entender lo que realmente ocurre dentro del procesador cada vez que un programa llama a una función recursiva, algo que los lenguajes de alto nivel ocultan automáticamente.